
Abstract 
Immortal time bias is one of the most common and consequential errors in real world evidence 
studies, and it almost always makes a treatment look more protective than it truly is. It arises 
whenever person time between cohort entry and the assignment of exposure is misclassified as 
exposed, or whenever exposure is defined using information that requires a patient to survive 
into the future. Because the bias is built into the structure of the analytic dataset rather than into 
how the data is analyzed, it cannot be removed by covariate adjustment.  

This paper provides a practical framework in SAS for recognizing immortal time bias at the 
design stage and removing it at the analysis stage. Using a fully reproducible simulated cohort in 
which the true treatment effect is null, we first reproduce the spurious protective hazard ratio that 
a naive time fixed Cox model returns, and we show how the same bias is already visible in the 
Kaplan-Meier curves. We then demonstrate three corrections in SAS: a time varying exposure 
model built with a counting process data layout and with a PROC PHREG programming 
statement, a landmark analysis, and a target trial emulation using the clone, censor, and weight 
approach. Each correction recovers the correct null result. The paper closes with a detection 
checklist so that attendees can audit their own studies for this type of bias. 

 

Keywords 

Immortal time bias, real world evidence, survival analysis, time varying exposure, landmark 
analysis, target trial emulation  



Introduction and Motivation 
 
Real world evidence now regularly informs regulatory decisions, clinical guidelines, payer 
policies, and many other aspects of healthcare. Much of that evidence comes from observational 
analysis of electronic health records and administrative claims, where treatment is not 
randomized but is observed as it unfolds over time. However, as the reliance on observational 
data grows, so does the risk of systematic errors that can lead to misleading conclusions. 
 
Among the most pervasive of these errors is immortal time bias, a structural flaw in study design 
or analysis that can falsely exaggerate the protective effects of a treatment. This bias occurs 
when there is a period of follow-up during which the outcome of interest cannot occur by design, 
typically because the patient must survive long enough to receive the treatment or meet the 
exposure criteria. In many real-world evidence studies, this "immortal" period is misclassified as 
exposed time, or the study structure inherently requires a patient to survive into the future to be 
included in a specific group. Because this bias is rooted in the temporal structure of the data 
rather than simple confounding, it remains resistant to standard covariate adjustment methods 
and requires specific methodological corrections at the design or analysis stage. 
 
A classic example of this bias comes from outside medicine. A widely cited study reported that 
Academy Award winning actors lived several years longer than their less celebrated peers. A 
later reanalysis showed that most of the apparent advantage disappeared once winning was 
treated as a status that accrues over time rather than as a fixed trait present from birth. The logic 
is simple: an actor must survive long enough to win, so the years lived before any win cannot 
fairly be credited to winning.  
 
The common thread is that exposure is defined by an event that happens after follow up begins: 
an award is given, a prescription is filled, a procedure is performed, a target dose is reached, or a 
laboratory response is achieved. Whenever exposure status depends on something in the future, 
the analyst risks giving the exposed group credit for time during which they could not have had 
the event. This paper shows how to spot the problem before writing code and how to remove it 
afterward, in SAS. 
 
 
Anatomy of Immortal Time Bias 
 
For any time to event analysis, three moments must be clearly defined and aligned with one 
another. First is time zero, the moment at which follow up begins and the clock starts. The 
second is eligibility, the moment at which the study inclusion criteria are assessed. The third is 
treatment assignment, the moment at which a patient is placed into an exposure group. Immortal 



time bias arises when the treatment group assignment uses information that is only available after 
time zero, so that the exposed group is defined partly by having survived.  
 
Two mechanisms produce the bias, and they often appear together. The first is misclassification 
of person time. When exposure is recorded as a fixed baseline attribute, such as ever treated 
versus never treated, the interval between time zero and the actual start of treatment is counted as 
exposed time even though no treatment had yet been received. That interval is immortal for the 
exposed group because a death during it would have removed the patient from the exposed 
category. The second mechanism is survivor selection. Requiring a patient to reach a treatment 
defining event in order to enter the exposed group selects for patients who lived long enough to 
get there, and correspondingly leaves shorter lived patients in the comparison group. 
 
Two properties of the bias are worth emphasizing. Its direction is predictable: it makes the 
exposure look protective, because the exposed group has been handed event free time. Its 
magnitude grows with the length of the immortal interval relative to total follow up and with the 
underlying event rate. This bias cannot be removed by adjustment. Confounding is about who 
receives treatment and is addressed with covariates or weights. Immortal time bias is about when 
person time is counted and to whom it is attributed. No baseline covariate encodes that error, so 
the fix must be structural. 
 
 
Detecting Immortal Time Bias at the Design Stage 
 
The most reliable detection happens before any analysis is done by asking three questions of the 
study design: 
 

1.​ When does follow up start? 
2.​ When is eligibility assessed? 
3.​ When is exposure assigned, and does that assignment depend on anything that happens 

after time zero? 
 
If the answer to the last question is yes, immortal time bias is almost certainly present, and the 
only question is how large it is. There are several recurring patterns that are red flags for 
immortal time bias. These include: 
 

●​ Exposure is defined as ever used during follow up and is then treated as a fixed 
characteristic measured at baseline. 

●​ Group assignment depends on a post baseline event such as a dispensed prescription, a 
completed procedure, an achieved dose, or a documented response. 



●​ Follow up for the exposed group is started at time zero even though the exposure itself 
began later. 

●​ A waiting period, run in, or new user window is built into the exposure definition in a 
way that requires survival. 

 
The common thread across every red flag is timing rather than covariates, which is why no 
amount of adjustment will rescue a study that fails these checks. A useful way to unify these 
checks is to articulate the target trial the study is emulating: the hypothetical randomized 
experiment with explicit eligibility criteria, a single time zero, and treatment strategies assigned 
at that moment. If you cannot place a patient into a strategy at one well defined time zero without 
consulting information from the future, the emulation contains immortal time.  
 
 
Simulated Cohort 
 
The rest of this paper demonstrates three techniques to correct against immortal time bias. To 
make the assessment of bias unmistakable, a simulated cohort of 3,000 patients is used in which 
the true treatment effect is null: the hazard of death is generated as a constant 0.15 per year 
regardless of treatment status. Treatment initiation is an independent process, with patients who 
survive long enough facing a constant hazard of 0.45 per year of starting treatment. 
Administrative censoring occurs at 5 years. Because death and treatment initiation are simulated 
as unrelated processes, any departure from a hazard ratio of one in the analyses that follow is 
known to be bias rather than signal. 
 

SAS Snippet 1. Simulated cohort construction 
/* ============================================================ 

   Simulated cohort: true treatment effect is null BY DESIGN. 

 ​ N = 3000 patients 

 ​ lambda_death = 0.15 / year   (constant hazard, independent of Rx) 

 ​ lambda_treat = 0.45 / year   (hazard of initiating treatment) 

 ​ administrative censoring at 5 years 

   Because t_death is generated with NO dependence on treatment 

   status, any hazard ratio different from 1 in later steps is 

   attributable entirely to bias, not to a real treatment effect. 

   ============================================================ */ 

data cohort; 

   call streaminit(20260810); 

   lambda_death  = 0.15; 

   lambda_treat  = 0.45; 

   admin_censor  = 5; 

  

   do id = 1 to 3000; 

  ​ t_death    ​ = rand('EXPONENTIAL') / lambda_death; 

  ​ t_treat_latent = rand('EXPONENTIAL') / lambda_treat; 



  

  ​ t_obs = min(t_death, admin_censor); 

  ​ event = (t_death <= admin_censor); 

  

  ​ /* a patient is only actually treated if the latent time to 

     ​initiation occurs before death / administrative censoring */ 
  ​ if t_treat_latent < t_obs then do; 

     ​initiates = 1; 
     ​t_treat   = t_treat_latent; 
  ​ end; 

  ​ else do; 

     ​initiates = 0; 
     ​t_treat   = .; 
  ​ end; 

  

  ​ output; 

   end; 

   keep id t_obs event initiates t_treat; 

run; 

 

In this dataset, of the 3,000 simulated patients, 1,549 (51.6%) experienced the event (death) by 5 
years and 2,146 (71.5%) initiated treatment at some point before death or administrative 
censoring, with a mean time to initiation of 1.45 years among initiators. 

 

Reproducing the Bias 
 
A naive analysis of this data treats the ever-treated indicator as a baseline covariate and counts 
all follow up time, including the immortal interval before initiation, as belonging to the treated 
group. We first plot the Kaplan-Meier curves by ever treated status and then fit a time fixed Cox 
model. 
 

SAS Snippet 2. Naive Kaplan-Meier curves and Cox model 
/* ------------------------------------------------------------ 

   Naive analysis: "ever treated" is coded as a FIXED baseline 

   covariate, so the interval between cohort entry and the actual 

   start of treatment is counted as exposed time for anyone who 

   eventually initiates. This is the immortal-time error. 

   ------------------------------------------------------------ */ 

proc lifetest data=cohort plots=survival(atrisk); 

   time t_obs*event(0); 

   strata initiates; 

   title "Naive Kaplan-Meier Curves by Ever-Treated Status"; 

run; 

  

proc phreg data=cohort; 

   model t_obs*event(0) = initiates / rl; 



   hazardratio initiates; 

   title "Naive Cox Model: Ever-Treated as a Fixed Baseline Covariate"; 

run; 

 

Figure 1. Naive Kaplan-Meier curves by ever-treated status. The treated curve cannot fall during the early follow-up 
period because patients who died early were never allowed into the treated group, a visible representation of the 

immortal time bias. 

Table 1. Analysis of Maximum Likelihood Estimates (naive model) 

Parameter DF Estimate StdErr ChiSq Pr>ChiSq Hazard Ratio 95% HR CL 

initiates 1 -1.484 0.052 816.76 <.0001 0.227 0.205 - 0.251 

 
The survival curves separate dramatically, with the treated group appearing to do far better, and 
the naive Cox model returns a hazard ratio of 0.23 (95% CI: 0.20-0.25). This can be interpreted 
as the treatment being associated with a 77 percent reduction in the hazard of death, which, as we 
know from the construction of the dataset, is entirely a result of bias. The separation in the 
survival curves gives a visual demonstration of the immortal time. 
 
 
Correction A: Time Varying Exposure 
The first and most general method to correct for this bias lets exposure change its value over 
time. Each patient contributes one interval while untreated and, if they initiate, a second interval 
while treated. The event is attached only to the final interval. Person time before initiation is then 
correctly counted as untreated, which removes the misclassification at the root of the bias. This 
can be implemented in SAS with a counting process data layout that records a start time, a stop 
time, and an event indicator for each interval. 
 



SAS Snippet 3. Time-varying exposure Cox model 
/* ------------------------------------------------------------ 

   Correction A: time-varying exposure via a counting-process 

   layout. Each patient who initiates contributes two rows: an 

   untreated interval (0, t_treat] and a treated interval 

   (t_treat, t_obs]. The event indicator is attached only to the 

   final interval. Person-time before initiation is now correctly 

   attributed to the untreated state. 

   ------------------------------------------------------------ */ 

data tv; 

   set cohort; 

   length start stop tvevent x 8; 

   if initiates = 1 then do; 

  ​ /* untreated interval */ 

  ​ start = 0;    ​stop = t_treat;  tvevent = 0;  ​ x = 0; 

  ​ if stop > start then output; 

  ​ /* treated interval */ 

  ​ start = t_treat;  stop = t_obs;​ tvevent = event;  x = 1; 

  ​ if stop > start then output; 

   end; 

   else do; 

  ​ start = 0;  stop = t_obs;  tvevent = event;  x = 0; 

  ​ output; 

   end; 

   keep id start stop tvevent x; 

run; 

  

proc phreg data=tv; 

   model (start,stop)*tvevent(0) = x / rl; 

   id id; 

   hazardratio x; 

   title "Time-Varying Exposure Cox Model (Counting-Process Layout)"; 

run; 

Table 2. Analysis of Maximum Likelihood Estimates (time-varying exposure model) 

Parameter DF Estimate StdErr ChiSq Pr>ChiSq Hazard Ratio 95% HR CL 

x (treated) 1 0.038 0.060 0.41 0.5218 1.039 0.924 - 1.168 

 
Fit on 5,146 intervals (3,000 patients contributing 6,146 candidate rows before dropping 
zero-length intervals for immediate initiators) and the same 1,549 events, the time-varying model 
returns a hazard ratio of 1.04 (95% CI: 0.92-1.17), correctly recovering the null.  
 
 
Correction B: Landmark Analysis 
 
A second method to address this bias is landmark analysis. This works by choosing a landmark 
time, keeping only patients still alive and under follow up at that point, classifying exposure by 



whether initiation occurred by the landmark, and then starting the clock at the landmark. Because 
every patient in the analysis has by definition survived to the landmark, no group is handed extra 
event free time. 

SAS Snippet 4. Landmark analysis Cox model 
/* ------------------------------------------------------------ 

   Correction B: landmark analysis. Keep only patients alive and 

   under follow-up at the landmark time L, classify exposure by 

   whether initiation occurred on or before L, then start the 

   clock at L using PHREG's (entry, exit) syntax so that no 

   patient is credited with event-free time before the landmark. 

   ------------------------------------------------------------ */ 

%let landmark = 1;   /* years; rerun with 2 as a sensitivity check */ 

  

data land; 

   set cohort; 

   where t_obs > &landmark; 

   exposed = (initiates = 1 and t_treat <= &landmark); 

   entry   = &landmark; 

run; 

  

proc phreg data=land; 

   model (entry,t_obs)*event(0) = exposed / rl; 

   hazardratio exposed; 

   title "Landmark Analysis Cox Model (Landmark = &landmark year)"; 

run; 

Table 3. Analysis of Maximum Likelihood Estimates (landmark models) 

Landmark N Events Estimate StdErr Hazard Ratio 95% HR CL 

L = 1 year 2,584 1,133 -0.001 0.063 0.999 0.884 - 1.129 

L = 2 years 2,208 757 -0.092 0.073 0.912 0.790 - 1.053 

 
 
At a one-year landmark, the model returns a hazard ratio of 1.00 (95% CI: 0.88-1.13); at a 
two-year landmark, the hazard ratio is 0.91 (95% CI: 0.79-1.05). Both confidence intervals span 
the null. As expected, the two-year landmark is less precise, since fewer patients survive to 
contribute person-time, illustrating the method's efficiency cost: 2,208 of the original 3,000 
patients remain at the two-year landmark versus 2,584 at one year, and the number of events falls 
from 1,133 to 757. 

The method is easy to explain and robust, but it carries two costs. It discards events and person 
time before the landmark, which reduces efficiency, and the result can depend on the landmark 
chosen, so reporting more than one landmark as a sensitivity check is good practice. Landmark 
analysis also classifies exposure once, at the landmark, so it is best suited to a one time exposure 
status rather than to treatments that start and stop repeatedly. 



 
Correction C: Target Trial Emulation  
 
The third method, and most advanced, is target trial emulation. The target trial framework 
removes immortal time by design rather than repairing it after the fact. We specify the trial we 
wish we could run and then emulate it. Here the eligibility criterion is cohort entry, time zero is 
cohort entry, and the two strategies being compared are to initiate treatment within a grace period 
and to refrain from initiating during that period. Because the strategies are defined at time zero, 
there is no window in which membership in a group requires survival. 
 
The challenge is that at time zero we do not yet know which strategy a patient will follow. The 
clone, censor, and weight approach resolves this by copying each patient into both strategies at 
time zero, then artificially censoring a copy at the moment the patient's observed behavior first 
becomes incompatible with the assigned strategy. Because that censoring is informative, inverse 
probability of censoring weights restore the lost person time.  
 

SAS Snippet 5. Target trial emulation: clone, censor, and weight 
/* ------------------------------------------------------------ 

   Correction C: target trial emulation via clone-censor-weight. 

   Grace period G = 1 year. Every patient is cloned into two 

   strategies at time zero: 

 ​ Clone A = "initiate treatment within the grace period" 

 ​ Clone B = "refrain from initiating during the grace period" 

   Each clone is artificially censored the moment the patient's 

   OBSERVED behavior first becomes incompatible with its assigned 

   strategy. That censoring is informative, so inverse-probability- 

   of-censoring weights (IPCW) are used to restore the lost 

   person-time in the final weighted Cox model. 

   ------------------------------------------------------------ */ 

%let grace = 1; 

  

data clone_a clone_b; 

   set cohort; 

  

   /* ---- Clone A: initiate-within-grace-period strategy ---- */ 

   x = 1; 

   if t_obs <= &grace then do; 

  ​ stop = t_obs;  tvevent = event;  deviate = 0;          ​ /* died/censored before G: 

never had the chance to deviate */ 

   end; 

   else if initiates = 1 and t_treat <= &grace then do; 

  ​ stop = t_obs;  tvevent = event;  deviate = 0;          ​ /* complied: initiated within 

the grace period */ 

   end; 

   else do; 

  ​ stop = &grace; tvevent = 0;  ​deviate = 1;          ​ /* deviated: survived past G 

without initiating */ 



   end; 

   output clone_a; 

  

   /* ---- Clone B: refrain-during-grace-period strategy ---- */ 

   x = 0; 

   if t_obs <= &grace and not(initiates = 1 and t_treat <= t_obs) then do; 

  ​ stop = t_obs;  tvevent = event;  deviate = 0;          ​ ​
            /* died/censored before G, never initiated */ 

   end; 

   else if initiates = 1 and t_treat <= &grace then do; 

  ​ stop = t_treat; tvevent = 0; ​deviate = 1;          ​ ​
            /* deviated: initiated during the grace period */ 

   end; 

   else do; 

  ​ stop = t_obs;  tvevent = event;  deviate = 0;          ​ ​
            /* complied throughout the grace period */ 

   end; 

   output clone_b; 

  

   keep id x stop tvevent deviate; 

run; 

  

data cloned; 

   set clone_a clone_b; 

   start = 0; 

run; 

  

/* Step 1: estimate, within each arm, the survival function of 

   remaining on-strategy (time to the artificial deviation-censoring 

   event), stratified by clone. */ 

proc phreg data=cloned noprint; 

   model stop*deviate(0) = ; 

   strata x; 

   baseline out=surv_dev survival=s_dev / method=pl; 

run; 

  

/* Step 2: attach each clone's last-known on-strategy survival 

   probability and invert it into a stabilized IPCW weight.    ​*/ 
proc sort data=cloned;   by x stop; run; 

proc sort data=surv_dev; by x stop; run; 

  

data weighted; 

   merge cloned(in=a) surv_dev(in=b); 

   by x stop; 

   retain s_last 1; 

   if b then s_last = s_dev; 

   ipcw = 1 / s_last; 

   if a; 

run; 

  

/* Step 3: fit the outcome model on the pooled, weighted clones.   */ 

proc phreg data=weighted covs(aggregate); 

   model stop*tvevent(0) = x / rl; 



   weight ipcw; 

   id id; 

   hazardratio x; 

   title "Target Trial Emulation: Clone-Censor-Weight Cox Model"; 

run; 

Table 4. Analysis of Maximum Likelihood Estimates (target trial emulation) 

Model Estimate StdErr ChiSq Pr>ChiSq Hazard Ratio 95% HR CL 

Clone-censor, 

unweighted 

0.005 0.048 0.01 0.9248 1.005 0.915 - 1.103 

Clone-censor-weight 

(IPCW) 

-0.076 0.036 4.46 0.0347 0.927 0.863 - 0.995 

  

Of the 3,000 clones assigned to the initiator strategy, 1,689 deviated by surviving past the grace 
period without initiating; of the 3,000 assigned to the non-initiator strategy, 977 deviated by 
initiating during the grace period. Before weighting, the pooled clone-censor model already lands 
almost exactly on the null (HR 1.00), because most of the correction in this design comes from 
correctly anchoring exposure strategies at time zero rather than from the weights themselves. 
After applying the stabilized inverse-probability-of-censoring weights, the estimate is 0.93 (95% 
CI: 0.86-1.00), which encompasses the null. 

 
 
Comparing the Methods 
 
Table 5 summarizes the estimates from the naive model and each of the corrections while Figure 
1 shows a forest plot of the hazard ratios obtained from each method. 
 

Table 5. Hazard ratio for the treatment effect by analytic method 

Method Hazard Ratio (95% CI) Recovers the Null? 

Naive 0.23 (0.20 - 0.25) No 

Time-Varying Exposure 1.04 (0.92 - 1.17) Yes 

Landmark Analysis (L=1y) 1.00 (0.88 - 1.13) Yes 

Target Trial Emulation 0.93 (0.86 - 1.00) Yes 

 



 
Figure 1. Estimated hazard ratio and 95% confidence interval by analytic method. The naive estimate sits far below 

the null while all three corrections cluster tightly around it. 

 
All three corrections recover the truth, so the choice among them rests on the question and the 
data. Time varying exposure is an efficient default method as it uses all person time, fits the 
standard counting process machinery, and handles exposures that turn on at a point in time. 
Landmark analysis is the simplest to explain and a useful sensitivity analysis, at the cost of 
discarding early follow up. Target trial emulation is the most principled when the design involves 
a grace period, when time zero is otherwise ambiguous, or when time varying confounding must 
be addressed alongside the immortal time, and it is the natural framework when emulating a 
specific trial. However, it is also the most sensitive to how its weights are estimated. 

 
 
Detection Checklist 
 
This short checklist can be used to quickly assess whether immortal time bias is present in a 
given study design: 

□   Time zero, eligibility, and treatment assignment are all defined at the same moment. 
□   Exposure status at time zero does not depend on any event that occurs after time zero. 
□   Person time before treatment actually begins is not counted as treated time. 
□   No patient is required to survive to a later event in order to enter the exposed group. 



□   The analysis can be stated as the emulation of a target trial with a single well defined 
randomization time. 

□   If exposure changes over follow up, it is modeled as time varying rather than as a fixed 
baseline trait. 

 
 
Limitations and Practical Notes 
The simulation isolates immortal time by removing confounding, but real studies face both 
problems at once. In practice the design corrections shown here are combined with adjustment 
for measured confounders, for example by adding covariates to the counting process model or by 
combining inverse probability of treatment weights with a time varying exposure or a full target 
trial emulation.  

Several technical details also deserve attention. Tied event times should be handled with the 
Efron approximation, which is the PROC PHREG default. Zero length intervals must be 
removed before fitting a counting process model, and patients who initiate at time zero need a 
consistent rule. Landmark results should be examined at more than one landmark, as illustrated 
above. 

The weights in a target trial emulation require careful estimation, stabilization, and a check that 
positivity holds across follow-up. The weight-construction step shown in Program 5 uses a 
simplified estimator: each clone's weight is fixed at the on-strategy survival probability observed 
nearest its own stop time, rather than a fully time-varying weight updated continuously across 
follow-up, which would be the most ideal analytic method. 

Finally, immortal time bias should be distinguished from related issues. It is not the same as time 
varying confounding, in which a time dependent covariate both responds to and predicts 
treatment, although the two often coexist and the weighting tools overlap. It is also distinct from 
depletion of susceptibles and from prevalent user bias, each of which has its own remedy. When 
the outcome is not death, competing risks may require cause specific or subdistribution hazard 
models, but the immortal time logic and its corrections apply unchanged. 

 
 
Conclusion 
Immortal time bias is common, it has a predictable direction that flatters the exposure, and it can 
be large enough to manufacture a 76 percent reduction in hazard out of a treatment with no effect 
at all. It is also avoidable. Because the bias lives in the structure of the analytic dataset, the cure 
is a programming discipline rather than a modeling trick: never classify a post baseline exposure 
as a fixed baseline trait, define time zero so that no group is handed event free time, and let 
exposure vary with time when it varies in reality.  

